Module 20.4: nag_f_dist

Probabilities and Deviate for an F-distribution

`nag_f_dist` provides procedures for computing tail probabilities and the deviate for an F- (or variance-ratio) distribution.

Contents

Procedure: nag_f_prob

1 Description

nag_f_prob returns the lower or upper tail probability for an F-distribution with ν_1 and ν_2 degrees of freedom.

2 Usage

```
USE nag_f_dist
[value =] nag_f_prob(tail, f, df1, df2  [, optional arguments])
```

The function result is a scalar of type real(kind=*wp*).

3 Arguments

3.1 Mandatory Arguments

tail — character(len=1), intent(in)

Input: the type of tail probability to be returned:

if tail = 'L' or 'l', the lower tail probability is returned;

if tail = 'U' or 'u', the upper tail probability is returned.

Constraints: tail = 'L', 'l', 'U' or 'u'.

f — real(kind=*wp*), intent(in)

Input: the value of the F-variate.

Constraints: f ≥ 0.0.

df1 — real(kind=*wp*), intent(in)

Input: the degrees of freedom, ν_1, for the variance of the numerator.

Constraints: df1 > 0.0.

df2 — real(kind=*wp*), intent(in)

Input: the degrees of freedom, ν_2, for the variance of the denominator.

Constraints: df2 > 0.0.

3.2 Optional Argument

error — type(nag_error), intent(inout), optional

The NAG *fl*90 error-handling argument. See the Essential Introduction, or the module document nag_error_handling (1.2). You are recommended to omit this argument if you are unsure how to use it. If this argument is supplied, it *must* be initialized by a call to nag_set_error before this procedure is called.

4 Error Codes

Fatal errors (error%level = 3):

error%code	Description
301	An input argument has an invalid value.

Warnings (error%level = 1):

error%code	**Description**
101	The value of `f` is too far out into the tails for the probability to be evaluated exactly. The result tends to approach 1 if `f` is large or 0 if `f` is small. However, the result returned is a good approximation to the required solution.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Mathematical Background

Let f be an F-distributed variate with ν_1 and ν_2 degrees of freedom. The lower tail probability $P(F \leq f : \nu_1, \nu_2)$ is defined by

$$P(F \leq f : \nu_1, \nu_2) = \frac{\nu_1^{\nu_1/2} \nu_2^{\nu_2/2} \Gamma\left((\nu_1 + \nu_2)/2\right)}{\Gamma(\nu_1/2)\Gamma(\nu_2/2)} \int_0^f F^{(\nu_1-2)/2} (\nu_1 F + \nu_2)^{-(\nu_1+\nu_2)/2} dF,$$

for ν_1, $\nu_2 > 0$, $f \geq 0$.

6.2 Algorithmic Detail

The probability is computed by means of a transformation to a beta distribution, $P_\beta(B \leq \beta : a, b)$:

$$P(F \leq f : \nu_1, \nu_2) = P_\beta \left(B \leq \frac{\nu_1 f}{\nu_1 f + \nu_2} : \frac{\nu_1}{2}, \frac{\nu_2}{2} \right).$$

For very large values of ν_1 and ν_2, greater than 10^5, a normal approximation is used. If only one of ν_1 or ν_2 is greater than 10^5 then a χ^2-approximation is used, see Abramowitz and Stegun [1].

6.3 Accuracy

The result is accurate to 5 significant digits.

Procedure: nag_f_deviate

1 Description

nag_f_deviate returns the deviate associated with the lower tail probability of an F-distribution with ν_1 and ν_2 degrees of freedom.

2 Usage

```
USE nag_f_dist
[value =] nag_f_deviate(p, df1, df2,  [, optional arguments])
```

The function result is a scalar of type real(kind=*wp*).

3 Arguments

3.1 Mandatory Arguments

p — real(kind=*wp*), intent(in)

Input: the lower tail probability of the F-distribution.

Constraints: $0.0 \leq$ p < 1.0.

df1 — real(kind=*wp*), intent(in)

Input: the degrees of freedom, ν_1, for the variance of the numerator.

Constraints: df1 > 0.0.

df2 — real(kind=*wp*), intent(in)

Input: the degrees of freedom, ν_2, for the variance of the denominator.

Constraints: df2 > 0.0.

3.2 Optional Argument

error — type(nag_error), intent(inout), optional

The NAG *fl*90 error-handling argument. See the Essential Introduction, or the module document nag_error_handling (1.2). You are recommended to omit this argument if you are unsure how to use it. If this argument is supplied, it *must* be initialized by a call to nag_set_error before this procedure is called.

4 Error Codes

Fatal errors (error%level = 3):

error%code	Description
301	An input argument has an invalid value.

Failures (error%level = 2):

error%code	Description
201	The value of p is too close to 0 or 1 for f_p to be computed. You should try another value for p.

Warnings (error%level = 1):

error%code	Description
101	The solution has failed to converge. But the result returned is a reasonable approximation.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Mathematical Background

Given the lower tail probability p of an F-distribution with ν_1 and ν_2 degrees of freedom, the deviate f_p associated with p is defined as the solution to

$$P(F \leq f_p : \nu_1, \nu_2) = p = \frac{\nu_1^{\nu_1/2} \nu_1 \nu_2^{\nu_2/2} \nu_2 \Gamma((\nu_1 + \nu_2)/2)}{\Gamma(\nu_1/2)\Gamma(\nu_2/2)} \int_0^{f_p} F^{(\nu_1-2)/2} (\nu_2 + \nu_1 F)^{(\nu_1+\nu_2)/2} dF,$$

where $\nu_1, \nu_2 > 0$; $0 \leq f_p < \infty$.

6.2 Algorithmic Detail

The value of f_p is computed by means of a transformation to a beta distribution,
$P_\beta(B \leq \beta : a, b)$:

$$P(F \leq f : \nu_1, \nu_2) = P_\beta \left(B \leq \frac{\nu_1 f}{\nu_1 f + \nu_2} : \frac{\nu_1}{2}, \frac{\nu_2}{2} \right).$$

For very large values of ν_1 and ν_2, greater than 10^5, a normal approximation is used. If only one of ν_1 or ν_2 is greater than 10^5 then a χ^2-approximation is used, see Abramowitz and Stegun [1].

6.3 Accuracy

The result is accurate to 5 significant digits.

Example 1: Calculation of probabilities and the deviate for an F-distribution

This example program shows how `nag_f_prob` returns the lower tail probability and upper tail probability for an F-distribution with ν_1 and ν_2 degrees of freedom. It also shows how `nag_f_deviate` calculates the deviate (f_calculated) associated with a given tail probability.

1 Program Text

Note. The listing of the example program presented below is double precision. Single precision users are referred to Section 5.2 of the Essential Introduction for further information.

```
    PROGRAM nag_f_dist_ex01

      ! Example Program Text for nag_f_dist
      ! NAG fl90, Release 3. NAG Copyright 1997.

      ! .. Use Statements ..
      USE nag_examples_io, ONLY : nag_std_out, nag_std_in
      USE nag_f_dist, ONLY : nag_f_prob, nag_f_deviate
      ! .. Implicit None Statement ..
      IMPLICIT NONE
      ! .. Intrinsic Functions ..
      INTRINSIC KIND
      ! .. Parameters ..
      INTEGER, PARAMETER :: wp = KIND(1.0D0)
      ! .. Local Scalars ..
      REAL (wp) :: df1, df2, f, f_calculated, prob, probl
      CHARACTER (1) :: tail
      ! .. Executable Statements ..

      WRITE (nag_std_out,*) 'Example Program Results for nag_f_dist_ex01'

      READ (nag_std_in,*)          ! Skip heading in data file

      WRITE (nag_std_out,*)
      WRITE (nag_std_out,*) 'TAIL    F      DF1     DF2      PROB     DEVIATE'
      WRITE (nag_std_out,*)

      DO
        READ (nag_std_in,*,end=20) tail, f, df1, df2

        prob = nag_f_prob(tail,f,df1,df2)

        probl = prob
        IF (tail=='u' .OR. tail=='U') probl = 1.0_wp - prob

        f_calculated = nag_f_deviate(probl,df1,df2)

        WRITE (nag_std_out,'(2x,A1,3x,F6.3,2F8.3,2F10.4)') tail, f, df1, df2, &
         prob, f_calculated
      END DO
20    CONTINUE

    END PROGRAM nag_f_dist_ex01
```

2 Program Data

```
Example Program Data for nag_f_dist_ex01
 'L'   5.5    1.5   25.5     :tail, f, df1, df2
 'U'  39.9    1.0    1.0
 'L'   2.5   20.25   1.0
```

3 Program Results

```
 Example Program Results for nag_f_dist_ex01

 TAIL    F       DF1     DF2       PROB     DEVIATE


  L    5.500   1.500  25.500    0.9837    5.5000
  U   39.900   1.000   1.000    0.1000   39.9023
  L    2.500  20.250   1.000    0.5342    2.5000
```

Additional Examples

Not all example programs supplied with NAG *fl*90 appear in full in this module document. The following additional examples, associated with this module, are available.

`nag_f_dist_ex02`

Calculation of the deviate associated with the lower tail probability of an F-distribution with known degrees of freedom.

References

[1] Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* Dover Publications (3rd Edition)